1. IDENTIFICATION

1.1 Digital-8-22-U-Sym

1.2 Unsigned Decimal Print

1.3 June 7, 1965

2. ABSTRACT

This subroutine permits the typeout of the contents of a computer word as a 4-digit, positive, decimal integer.

3. REQUIREMENTS

3.1 Storage

This subroutine requires 38 core locations.

3.3 Equipment

Basic PDP-8 with ASR 33.

4. USAGE

4.1 Loading

The subroutine may be placed in core by use of the Binary Loader. See Digital-8-2-U-Rim for full details. The symbolic tape provided is either assembled with the user program or separately with the proper origin setting.

4.2 Calling Sequence

The subroutine is called by the usual JMS instruction with the number to be printed in the AC. Return to the location following that of the calling JMS.

5. RESTRICTIONS (Not Applicable)

6. DESCRIPTION

6.1 Discussion

This is a basic subroutine used to obtain decimal output corresponding to binary words in memory. The program operates in a straightforward manner. First the binary equivalent of 1000 is subtracted from the original number until a negative result is obtained. A count is kept of the number of subtractions necessary to accomplish this, thus yielding the most significant decimal digit. This process is repeated--using the proper power of ten, to give the three remaining decimal digits.

7. METHOD

7.1 Discussion

This method of binary to binary coded decimal conversion is compact and easily understood if it is not sophisticated. The latter consideration is of little consequence since the subroutine is output limited.

8. FORMAT

8.3 Output Data

Output is in the form of four consecutive decimal digits. No sign is printed. Spacing, tabulation, carriage return, etc. are not provided for in this subroutine. See Digital-8-19-U-Sym, which contains short subroutines for the latter purposes.

9. EXECUTION TIME

9.3 Average

This subroutine is output limited.

10. PROGRAM

10.4 Program Listing

```
                    /PDP-8 UNSIGNED DECIMAL PRINT
                    /CALL WITH NUMBER TO BE TYPED IN AC
                    /RETURN TO LOCATION FOLLOWING CALLING JMS
0200    0000        DECPRT,     0
0201    3243                    DCA VALUE       /SAVE INPUT
0202    3244                    DCA DIGIT       /CLEAR
0203    1235                    TAD CNTRZA
0204    3245                    DCA CNTRZB      /SET COUNTER TO FOUR
0205    1234                    TAD ADDRZA
0206    3213                    DCA ARROW       /SET TABLE POINTER
0207    7410                    SKP
0210    3243                    DCA VALUE       /SAVE
0211    7100                    CLL
0212    1243                    TAD VALUE
0213    1236        ARROW,      TAD TENPWR      /SUBTRACT POWER OF TEN
0214    7430                    SZL
0215    2244                    ISZ DIGIT       /DEVELOP BCD DIGIT
0216    7430                    SZL
0217    5210                    JMP ARROW-3     /LOOP
0220    7600                    CLA             /HAVE BCD DIGIT
0221    1244                    TAD DIGIT       /GET DIGIT
0222    1242                    TAD K260        /MAKE IT ASCII
0223    6041                    TSF             /OR = TAD DIGIT
0224    5223                    JMP .-1         /JMP TDIG: SEE 8-19-U-SYM
0225    6046                    TLS             /TYPE DIGIT
```

```
0226  7600          CLA
0227  3244          DCA DIGIT       /CLEAR
0230  2213          ISZ ARROW       /UPDATE POINTER
0231  2245          ISZ CNTRZB      /DONE ALL FOUR?
0232  5212          JMP ARROW-1     /NO: CONTINUE
0233  5600          JMP I DECPRT    /YES: EXIT
0234  1236  ADDRZA, TAD TENPWR
0235  7774  CNTRZA, -4
0236  6030  TENPWR, -1750          /ONE THOUSAND
0237  7634          -0144           /ONE HUNDRED
0240  7766          -0012           /TEN
0241  7777          -0001           /ONE
0242  0260  K260,   260
0243  0000  VALUE,  0
0244  0000  DIGIT,  0
0245  0000  CNTRZB, 0
            $
```

11. DIAGRAMS (Not Applicable)

12. REFERENCES

Digital-8-19-U-Sym. Teletype Output Subroutines